


Ejemplo de aliasing

Federico Miyara

Este ejemplo utiliza cuatro señales, una senoide, una onda cuadrada sin aliasing obtenida superponiendo armónicos hasta $F_s/2$, una onda cuadrada con poco aliasing obtenida superponiendo armónicos hasta $C_{alias} F_s/2$ donde $C_{alias} > 1$, y una onda cuadrada con máximo aliasing obtenida como el signo de una senoide. En todos los casos se generan armónicos impares que se reducen según $1/n$. Los que exceden de la frecuencia de Nyquist $F_s/2$ se van reflejando alternativamente en $F_s/2$ y 0.

En la figura se muestra conceptualmente este fenómeno, aunque para mayor claridad del dibujo decaen linealmente en lugar de con $1/n$.



f_0 es la frecuencia fundamental

k tiene un valor distinto para cada armónico o punto de reflexión en $F_s/2$ o en 0

d es el retraso de k para obtener el número de armónico no nulo

sg es el signo que se le aplica al incremento de frecuencia

El número de armónico (sólo armónicos impares en este ejemplo) se calcula como

$$n = 2(k - d) - 1$$

La frecuencia correspondiente al índice k (tanto para armónicos como para puntos de reflexión) es

$$f_{env}(k) = \text{mod}(d, 2) F_s/2 + sg (nf_0 - d F_s/2)$$

La envolvente de los armónicos y puntos de reflexión disminuye con $1/n$, por ser armónicos de una onda cuadrada. Para armónicos:

$$Z_{env}(k) = 4/\pi / n$$

Para puntos de reflexión:

$$Z_{\text{env}}(k) = 4/\pi / (d * F_s / 2 / f_o)$$

Los criterios para decidir si la envolvente se refleja hacia frecuencias crecientes o decrecientes son los siguientes. Para reflexión en $F_s/2$ se deben dar las condiciones:


$$\begin{aligned} n f_o &> (1+d) F_s / 2 \\ sg &= 1 \end{aligned}$$

Para reflexión en 0 se deben dar las condiciones:

$$\begin{aligned} n f_o &> (1+d) F_s / 2 \\ sg &= -1 \end{aligned}$$

En la figura siguiente se ven las señales en el tiempo y en el espectro. En la segunda fila se observa claramente el fenómeno de Gibbs (un sobrepico en las transiciones cuando el espectro es acotado). En la tercera el fenómeno se complica perdiendo regularidad a causa del aliasing. En la cuarta fila la onda cuadrada en apariencia es perfecta, pero el aliasing es máximo

Al final se detalla el script de Scilab usado para obtener estas figuras:



alias_vs_no_alias_no_spectral_leakage.sci

```
// Ejemplo de señal con aliasing
//
// Para visualizar el problema del aliasing se generan
// cuatro ondas: Senoidal, cuadrada sin aliasing, cuadrada
// con poco aliasing y cuadrada con mucho aliasing. La
// cuadrada sin aliasing se genera sumando armónicos hasta
//  $F_s/2$ , La cuadrada con poco aliasing, sumando armónicos
// que exceden la tasa de Nyquist  $F_s/2$ . En este caso los
// armónicos se "reflejan" alternativamente alrededor de
//  $F_s/2$  y 0. La onda cuadrada con mucho aliasing se logra
// tomando el signo de una senoide, por lo que el espectro
// exhibe infinito aliasing y, consecuentemente, infinitas
// reflexiones.
//
// Para evitar contaminar el ejemplo se usó una señal sin
// derrame espectral. Para ello se eligió la frecuencia
// de modo que quepa un número entero de periodos en la
// ventana de análisis:
//
//  $f_0 = m \cdot F_s / N$ 
//
// donde m es el número de línea de la FFT que corresponde
// a  $f_0$ 
//
// NOTAS
// 1) Reducir m es equivalente a aumentar la tasa de
// muestreo en relación con la frecuencia fundamental.
// Se puede observar que en la onda cuadrada ideal el
// aliasing se reduce conforme aumenta  $F_s/f_0$ , pero la
// única manera de eliminarlo del todo es limitar la
// banda.
// 2) Para mostrar los repliegues o reflexiones del
// espectro se traza en rojo la envolvente de los
// armónicos
//
// -----
// Federico Miyara
// 2024-10-24
// 2024-11-09

clear
// Tasa de muestreo
Fs = 44100
// Tamaño de la ventana para FFT
N = 4096

// Número de línea espectral de la FFT donde
// se encontrará la frecuencia fundamental  $f_0$ 
// (armónico 1)
m = 95
```

```

// Periodo de las ondas para que no haya derrame
// espectral (se consigue haciendo caber exactamente
// m periodos en la ventana de análisis FFT.
// NOTA: salvo que además m sea una potencia de 2,
// la señal digital obtenida no es periódica, aunque
// sí cumple el teorema de muestreo (no hay alias)
// representa una onda periódica en tiempo continuo
To = N/Fs/m
// Frecuencia fundamental
fo = 1/To;
// Duración de las señales
T = N/Fs;
// Vector de tiempos
t = (0:N-1)/Fs;

// Señal senoidal
x = sin(2*pi*fo*t);

// Onda cuadrada sin alias
y = 0;
for k=1:2:floor(Fs/2/fo)
    y = y + 4/pi * sin(2*pi*k*fo*t) / k;
end

// Onda cuadrada con alias a partir de su
// serie de Fourier con armónicos hasta una
// frecuencia Falias veces mayor que la de
// Nyquist (Fs/2).
// Inicialización de la señal
z = 0;
// Factor de aliasing
C_alias = 8/3
// Cantidad de armónicos
Narm = ceil(1 + floor(Fs/2/fo/2)*C_alias)
// Generación de la señal
for k=1:Narm
    z = z + 4/pi * sin(2*pi*(2*k-1)*fo*t) / (2*k-1);
end

// Onda cuadrada con alias generada
// aplicando la función signo a una
// senoide ligeramente desfasada para
// evitar valores 0.
w = sign(sin(2*pi*fo*t + 0.0001));

// Espectros
X = abs(fft(x(1:N))) * 2/N;
Y = abs(fft(y(1:N))) * 2/N;
Z = abs(fft(z(1:N))) * 2/N;
W = abs(fft(w(1:N))) * 2/N;
W = max(0.000001, W);

// Vector de frecuencias
f = (0:N/2-1)*Fs/N;

```

```

// Cálculo de la envolvente reflejada de los
// armónicos teóricos. Dado que queremos
// resaltar los puntos de reflexión, que se
// darán alternativamente en  $F_s/2$  y en 0,
// agregaremos al número de armónicos la
// cantidad de reflexiones.
N_reflex = floor((2*Narm-1)*fo/(Fs/2))
N_puntos_env = Narm + N_reflex
// Inicializar el índice k, la cantidad d de
// reflexiones hasta el momento y el signo
// sg con que se incrementará la frecuencia
// de los parciales reflejados (cuando se
// reflejan en  $F_s/2$ ,  $sg = -1$ , mientras que
// cuando se reflejan en 0,  $sg = 1$ )
k = 1
d = 0
sg = 1
clear fenv Zenv
while k <= N_puntos_env
    // Número de armónico no nulo
    n = 2*(k-d)-1;
    if n*fo > (1+d)*Fs/2 & sg==1
        // Signo con el que se incrementará la
        // frecuencia
        sg = -sg;
        // Número de reflexiones hasta el momento
        d = d+1;
        // Frecuencia del punto de reflexión hacia
        // la izquierda
        fenv(k) = Fs/2;
        Zenv(k) = 4/pi/(d*Fs/2/fo);
    elseif n*fo > (1+d)*Fs/2 & sg==-1
        // Signo con el que se incrementará la
        // frecuencia
        sg = -sg;
        // Número de reflexiones hasta el momento
        d = d+1;
        // Frecuencia del punto de reflexión hacia
        // la derecha
        fenv(k) = 0;
        Zenv(k) = 4/pi/(d*Fs/2/fo);
    else
        // Las reflexiones se dan alternativamente
        // en  $F_s/2$  (cuando d es impar) y en 0 (cuando
        // d es par)
        fenv(k) = modulo(d,2)*Fs/2 + sg*(n*fo - d*Fs/2);
        Zenv(k) = 4/pi/n;
    end
    k = k + 1;
end

// Seleccionar si el eje vertical será lineal
// o logarítmico
caso = "log"
select caso
case "lin"

```

```

    ejes = "nn";
case "log"
    ejes = "nl";
end

// Gráficas
scf(1);
clf(1);
gcf().figure_position = [200,100];
subplot(4,2,1)
plot(t(1:2*N/m), x(1:2*N/m));
gca().data_bounds = [0 -1.2; t(2*N/m) 1.2];
subplot(4,2,2)
plot(ejes, f, X(1:N/2))
gca().data_bounds = [0 0.001; Fs/2 4/%pi];
subplot(4,2,3)
plot(t(1:2*N/m), y(1:2*N/m));
gca().data_bounds = [0 -1.2; t(2*N/m) 1.2];
subplot(4,2,4)
plot(ejes, f, Y(1:N/2))
gca().data_bounds = [0 0.001; Fs/2 4/%pi];
subplot(4,2,5)
plot(t(1:2*N/m), z(1:2*N/m));
gca().data_bounds = [0 -1.2; t(2*N/m) 1.2];
subplot(4,2,6)
plot(ejes, f, Z(1:N/2), fenv, Zenv, "r")
gca().data_bounds = [0 0.001; Fs/2 4/%pi];
subplot(4,2,7)
plot(t(1:2*N/m), w(1:2*N/m));
gca().data_bounds = [0 -1.2; t(2*N/m) 1.2];
subplot(4,2,8)
plot(ejes, f, W(1:N/2))
gca().data_bounds = [0 0.001; Fs/2 4/%pi];

```